

Textul si imaginile din acest document sunt licentiate

Attribution-NonCommercial-NoDerivs

CC BY-NC-ND



Codul sursa din acest document este licentiat

Public-Domain

Esti liber sa distribui acest document prin orice mijloace consideri (email, publicare pe website / blog, printare, sau orice alt mijloc), atat timp cat nu aduci nici un fel de modificari acestuia. Codul sursa din acest document poate fi utilizat in orice fel de scop, de natura comerciala sau nu, fara nici un fel de limitari.

Arduino udă ghivecele cu flori

Se dau: o problemă cotidiană foarte des întâlnită și o mulțime de soluții parțiale ce nu rezolvă niciodată problema propriu-zisă. Cam așa se poate rezuma problema implementării unui sistem bazat pe placa de dezvoltare Arduino Uno ce permite automatizarea procesului de udare a ghivecelor cu flori. Majoritatea exemplarelor existente permit măsurarea umidității din sol dar nu implementează și partea de acționare propriu-zisă mulțumindu-se să ”te tragă de mânecă” când trebuie să uzi florile sau să-ți trimită un email de atenționare ☺. Lecția de față își propune să implementeze o soluție completă (achiziție plus acționare) ajutând astfel pe oricine să-și permită să uite să ude florile fără a le pune în pericol.



Componentele utilizate de sistem sunt foarte comune facilitând astfel implementarea la un cost rezonabil: placă de dezvoltare [Arduino Uno](#), [shield drivere de motoare L298](#), [senzor umiditate sol](#), [pompa de apă micro](#) și [ecran LCD Nokia](#) (opțional).





<https://www.robofun.ro/mecanice/mecanice-altele/pompa-apa-micro>

<https://www.robofun.ro/senzori/vreme/soil-moisture-sensor>

Shield-ul de motoare L298 este utilizat de obicei în construcția platformelor robotice telecomandate (a se vedea seria de roboți [FlexyBot](#)) dar se poate utiliza și în cazul nostru – pompa de apă nu este altceva decât un motor de curent continuu. Se poate utiliza și un releu pentru comanda pompei dar se pierde capacitatea de control al turăției (debitului). Shield-ul este capabil să comande două motoare, sistemul propus folosește o singură ieșire de comandă dar permite o extindere simplă ulterioară.


<https://www.robofun.ro/forum/>

Conectarea pompei de apă la shield se va face pe canalul Motor2 având grijă la pinul marcat cu bulină roșie al pompei. Shield-ul va fi utilizat cu jumper-ul de alimentare comună pus, astfel vom putea alimenta și placa de dezvoltare și pompa de la un singur [alimentator de 9V](#) conectat la placa de dezvoltare (mufa de alimentare a plăcii de dezvoltare). Dacă se respectă conectarea indicată sensul de udare al pompei va fi următorul:




Pentru conectarea pompei de apă la un recipient cu apă (pompa nu este proiectată să fie conectată la o sursă de apă cu presiune proprie!!!) și la ghivecele cu flori ce urmează a fi udate se poate folosi tub de silicon cu diametrul interior de 3mm (sunt utilizate de obicei la acvarii). În plus se pot utiliza diverși conectori specifici pentru a realiza o instalație care să permită udarea mai multor

ghivece simultan:



Este recomandat ca recipientul (rezervorul) de apă să nu fie mai jos ca sistemul de udare pentru a nu forța pompa să ridice apa.

Senzorul de umiditate sol (pinul SIG) se va conecta la placa de dezvoltare (prin intermediul shield-ului) la pinul analogic A0. Pinul de alimentare al senzorului se va conecta la pinul A2, senzorul va fi alimentat doar când se va efectua citirea. Alimentarea continuă a senzorului poate duce la deterioarea rapidă a acestuia din cauza efectului de electroliză. Ieșirea senzorului nu este calibrată, nu putem calcula precis cantitatea de apă din sol pe baza acestui senzor. Citirile sunt afectate destul de mult și de tipul de sol din ghiveciul cu flori. Având în vedere că citirile analogice se află în plaja 0 – 1023 vom stabili un prag sub care vom considera că ghiveciul are nevoie de apă (pragul stabilit în program va fi 500).



Este recomandată parcurgerea următorului material pentru a înțelege mai bine funcționarea senzorului de umiditate a solului:

Soil Moisture Sensor Hookup Guide

<https://learn.sparkfun.com/tutorials/soil-moisture-sensor-hookup-guide>


Ecranul LCD va permite urmărirea funcționării sistemului dar sistemul poate îndeplini sarcina stabilită și fără conectarea acestuia. Ecranul grafic LCD Nokia funcționează la 3.3V, conectarea linilor la pinii Arduino de 5V scade durata de viață a afișajului. Este recomandată adăugarea unui circuit de translatăre a nivelurilor logice, a se vedea și materialul:

Graphic LCD Hookup Guide

<https://learn.sparkfun.com/tutorials/graphic-lcd-hookup-guide>

Conexiunile necesare afișajului LCD sunt următoarele:

- Pinul LED al LCD-ului, prin intermediul unei rezistențe de 220ohm, se va conecta la 5V (sau un un pin PWM dacă se dorește variația intensității luminii de fundal);
- Pinul SCLK al LCD-ului se va conecta la pinul 13 al plăcii de dezvoltare;
- Pinu DIN (MOSI) al LCD-ului se va conecta la pinul 11 al plăcii de dezvoltare;
- Pinul D/C la pinul 8;
- Pinul RST la pinul 7;
- Pinul SCE la pinul 4;
- Pinul GND la GND;
- Pinul Vcc la 3.3V.



Pentru a putea lucra cu LCD-ul programul va utiliza bibliotecile [Adafruit GFX](#) și [Adafruit PCD8544](#).

```
#include <SPI.h>
#include <Adafruit_GFX.h>
#include <Adafruit_PCD8544.h>
Adafruit_PCD8544 display = Adafruit_PCD8544(8, 4, 7);
```

```
#define MOTOR2_PIN1 3
#define MOTOR2_PIN2 5
#define SOIL_PIN     A0
#define SOIL_VCC     A2
```


```
void setup() {  
    display.begin();  
    display.setContrast(50);  
    delay(1000);  
    display.clearDisplay();  
    display.display();  
    pinMode(MOTOR2_PIN1, OUTPUT);  
    pinMode(MOTOR2_PIN2, OUTPUT);  
    digitalWrite(MOTOR2_PIN2, LOW);  
    pinMode(SOIL_VCC, OUTPUT);  
    digitalWrite(SOIL_VCC, LOW);  
    delay(1000);  
}
```

În cadrul secțiunii *loop()* a programului se va efectua citirea senzorului de umiditate a solului (cu activarea prealabilă a pinul A2 pentru a alimenta senzorul) și în funcție de valoarea citită (dacă valoarea citită este mai mică de 500) se pornește un ciclu de udare format din trei perioade de 1 secundă alternate cu trei perioadă de pauză de 1 secundă. Pompa este comandată la jumătatea puterii maxime (125 din 255). În funcție de dimensiunea ghiveciului sau / și de numărul de ghivece udate ciclul de udare poate fi ajustat ca durată și putere comandată pompei de apă.

```
unsigned long lasttime = 0;  
  
void loop()  
{  
    int soilhumidity;  
    digitalWrite(SOIL_VCC,HIGH);  
    delay(1000);  
    soilhumidity = analogRead(SOIL_PIN);  
    digitalWrite(SOIL_VCC,LOW);  
    display.clearDisplay();  
    display.display();  
    display.setTextSize(1);
```

```
display.setTextColor(BLACK);
display.setCursor(0,0);
display.print("Soil hum:");
display.println(soilhumidity);
display.print("Time:");
display.println(millis()/1000);
if (soilhumidity<500) {
    display.println("Status: ON");
    display.display();
    delay(100);
    analogWrite(MOTOR2_PIN1, 125);
    delay(1000);
    analogWrite(MOTOR2_PIN1, 0);
    delay(1000);
    analogWrite(MOTOR2_PIN1, 125);
    delay(1000);
    analogWrite(MOTOR2_PIN1, 0);
    delay(1000);
    analogWrite(MOTOR2_PIN1, 125);
    delay(1000);
    analogWrite(MOTOR2_PIN1, 0);
    lasttime = millis()/1000;
}
else {
    display.println("Status: OFF");
}
display.print("LastTime:");
display.println(lasttime);
display.display();
delay(100);
}
```


Afişajul LCD va afişa în permanenţă valoarea citită de la senzorul de umiditate sol, timpul în secunde de la pornirea sistemului: Time, starea sistemului de comandă (ON/OFF – udă/nu udă) şi timpul la care s-a efectuat ultimul ciclu de udare: LastTime (estimat în secunde de la pornirea sistemului). Afişarea timpului poate fi rafinată prin transformarea valorii afişate în HH:MM:SS nu doar în secunde.



Programul a fost dezvoltat şi testat utilizând Arduino IDE 1.8.1 şi bibliotecile Adafruit GFX 1.1.8 şi Adafruit PCD8544 1.0.0.

Sistemul prezentat poate fi îmbunătăţit prin adăugarea unor elemente suplimentare de achiziţie (temperatură, umiditate aer, putere lumină solară) pentru a realiza un sistem de automatizare care să țină seama de mai mulţi factori în alegerea momentului optim de udare. Pentru inspiraţie se pot studia şi următoarele proiecte:

Arduino Automatic Watering System for Plants Sprinkler

<http://www.instructables.com/id/Arduino-Automatic-Watering-System-For-Plants>

Automatic Plant Watering and Soil Moisture Sensing

<http://www.instructables.com/id/Automatic-Plant-Watering-and-Soil-Moisture-Sensing>

Self-Watering Plant

<http://www.instructables.com/id/Self-Watering-Plant>