

ROBOFUN.RO
LECȚIA XIX

CURS GRATUIT

ARDUINO ȘI ROBOTICĂ

**Identificare RFID
cu Arduino**

Textul si imaginile din acest document sunt licentiate

Attribution-NonCommercial-NoDerivs
CC BY-NC-ND



Codul sursa din acest document este licentiat

Public-Domain

Esti liber sa distribui acest document prin orice mijloace consideri (email, publicare pe website / blog, printare, sau orice alt mijloc), atat timp cat nu aduci nici un fel de modificari acestuia. Codul sursa din acest document poate fi utilizat in orice fel de scop, de natura comerciala sau nu, fara nici un fel de limitari.

Arduino UNO si cititorul RFID ID-12

In urmatorul tutorial vei utiliza un Arduino UNO si un cititor RFID pentru a citi diverse tag-uri care pot fi folosite la un sistem simplu de securitate sau la identificarea unor obiecte.

RFID este prescurtarea de la termenul Radio-Frequency Identification si este format din trei parti:

- o antena, care alimenteaza cu energie tag-urile si stabileste comunicatia dintre cititor si tag.
- cititorul, care interpreteaza datele.
- tag-ul, care are un cod unic.


Atunci cand tag-ul se afla la mai putin de 5 cm de antena acesta raspunde cu un cod unic format din 12 digiti. Cititorul preia codul unic si il transmite mai departe catre Arduino.


Pentru acest tutorial vei avea nevoie de urmatoarele componente:

- Arduino UNO
- led brick
- adaptor RFID pe USB
- cititor RFID ID-12
- diverse tag-uri
- un alimentator extern (12V)
- fire pentru conexiuni

Diagrama de conectare este cea din tabelul de mai jos.

Adaptor RFID Vcc	Arduino UNO 5V
Adaptor RFID Gnd	Arduino UNO Gnd
Adaptor RFID TX	Arduino UNO Digital 7
Led brick IN	Arduino UNO Digital 12
Led brick Gnd	Arduino UNO Gnd





Cum functioneaza.

Cititorul RFID ID-12 se conecteaza la adaptorul USB deoarece vei prelua semnalul TX si il vei duce acest semnal la pinul digital 7 al Arduino. Alimentarea adaptorului o vei face din Arduino (din pinii parcati cu 5V si GND). Daca cititorul detecteaza un tag, acesta va transmite adresa lui prin pinul TX catre Arduino.



Alimenteaza platforma Arduino UNO din alimentatorul extern. Daca vei apropia tagul RFID de cititor, acesta va produce un sunet scurt si se va aprinde led-ul Read ceea ce inseamna ca cititorul detecteaza corect tag-ul.

Poti utiliza multe tipuri de tag-uri (tip breloc, bratara, buton sau cartela, ca mai jos).



Codul sursa.

Incarca urmatorul sketch in Arduino.

```
#include <SoftwareSerial.h>
SoftwareSerial rfid(7, 6); // RX, TX


const int ledRosu = 12;
boolean stare = false;
int reading = 0;

void setup() {
  Serial.begin(9600); // initializeaza portul serial la 9600 baud
  rfid.begin(9600);   // initializeaza modulul rfid la 9600 baud
  pinMode(ledRosu, OUTPUT);
}

void loop() {
  while (rfid.available()) {
    reading = rfid.read(); // citeste byte-ul
    if (reading == 2) {
      stare = true; // '2' reprezinta inceput de string
      Serial.print("Serie RFID tag: ");
    }
    if (reading == 3) {
      stare = false; // '3' reprezinta sfarsit de string
      Serial.println("");
      digitalWrite(ledRosu, HIGH);
      delay(5000);
      digitalWrite(ledRosu, LOW);
    }
    if (stare && reading != 2 && reading != 10 && reading != 13)
  { // transmite fiecare byte la serial monitor
    Serial.write(reading);
  }
}
}
```

Dupa ce ai incarcat sketch-ul, deschide monitorul serial. Apropie diverse tag-uri de cititorul RFID. Vei auzi un beep scurt, led-ul brick se va aprinde pentru 5 secunde iar monitorul serial va afisa seria unica a tagului.

Monitorul serial va indica cod-urile unice RFID ca mai jos.



În funcția `setup()` sunt initializate două porturi seriale. Primul este folosit de Arduino pentru transmiterea informației către calculator (serial hardware), iar cel de-al doilea este o emulare de comunicare serial pe pinul digital 7 (software serial). În acest mod, nu vei fi obligat să deconectezi firul TX de fiecare dată când încarci sketch-ul.

Bucula `loop()` monitorizează apariția adreselor RFID. Dacă apare o adresă, ea va începe întotdeauna cu caracterul „2” și se va sfârși cu caracterul „3”. Monitorul serial va printa seria unică și checksum-ul codului. Atunci când s-a detectat digit-ul „3” înseamnă că nu mai urmează nici un caracter astfel că se printează o nouă linie la monitorul serial și se porneste led-ul timp de 5 secunde.

Incarca urmatorul sketch.

```
#include <SoftwareSerial.h>
SoftwareSerial rfid(7, 6); // RX, TX
const int ledRosu = 12;
boolean stare = false;
int reading = 0;
int contor=0;
char serieTag[13];
char serieUnica[13] = "290067647D57";
void setup() {
    Serial.begin(9600); // initializeaza portul serial la 9600 baud
    rfid.begin(9600); // initializeaza modulul rfid la 9600 baud
    pinMode(ledRosu, OUTPUT);
}

void loop() {
    while (rfid.available()) {
        reading = rfid.read(); // citeste byte-ul
        if (reading == 2) {
            stare = true; // '2' reprezinta inceput de string
            Serial.print("Serie RFID tag: ");
        }
        if (reading == 3) {
            stare = false; // '3' reprezinta sfarsit de string
            Serial.println("");
            contor = 0;
            if (strcmp(serieTag, serieUnica) == 0) {
                digitalWrite(ledRosu, HIGH);
                delay(5000);
                digitalWrite(ledRosu, LOW);
            }
        }
        if (stare && reading != 2 && reading != 10 && reading != 13)
        { // transmite fiecare byte la serial monitor
            serieTag[contor]=reading;
            Serial.write(reading);
            contor++;
        }
    }
}
```

Sketch-ul pe care l-ai incarcat functioneaza asemanator cu primul, diferenta fiind ca led-ul se va aprinde doar la un anumit tag, respectiv cel care are seria „290067647D57”.

In sketch s-a declarat o variabila numita contor, ce va tine cont de pozitia fiecarui caracter al adresei unice, si doua string-uri de tip char respectiv serieTag si serieUnica. In serieTag se va stoca adresa detectata de cititor iar seriaUnica reprezinta seria tagului ce va aprinde led-ul.

Prin functia strcmp(serieTag, serieUnica) == 0 se compara cele doua string-uri. Daca functia returneaza 0, inseamna ca exista egalitate (tag-ul va aprinde led-ul).